
Interfacing the AlphaGUARD Radon Monitor 
with Campbell Scientific's CR1000 Datalogger
Th is application note describes interfacing a CR1000 datalogger with a “standard version” of the AlphaGUARD 
Professional Radon Monitor Model P30 (not the “overdrive” version).  Th e eff orts to interface the AlphaGUARD 
Radon Monitor with the CR1000 Datalogger, and the writing of this application note were accomplished in col-
laboration with personnel from DRI, the environmental research arm of the Nevada System of Higher Education1. 

Th e AlphaGUARD is manufactured by Saphymo GmbH (www.saphymo.de)2.  Th is document is intended as a 
supplement to the manufacturer’s brochure and the manuals titled: “User Manual Portable Radon Monitor “Al-
phaGUARD dated 12/98” and “AlphaGUARD Data Transfer – Specifi cations version 4.0 dated 7/97”.   

1.  Radon Detection Basics
Two techniques are available for measuring radon activity; the best technique to use 
depends on the application.  The techniques are integrative, passive radon sampling and 
continuous active radon sampling.  Integrative, passive radon sampling does not require 
electrical power, whereas continuous active radon sampling does. 

Examples of sensors that can measure radon levels include alpha track detectors, acti-
vated charcoal adsorption devices, and the AlphaGUARD monitor.  Alpha track detectors 
are integrative, passive radon sampling devices that do not require ac power.  They 
contain a thin piece of plastic or film mounted in the detector.  Radon flows into the 
detector through a filtered opening.  As the radon inside the detector decays, the emitted 
alpha particles hit the film forming tracks on the film.  The tracks are counted to deter-
mine radon concentration.  These detectors are not very effective in measuring high 
concentrations of radon [Spe04].

Activated charcoal devices are passive sensors that do not require ac power.  They consist 
of a canister that holds granular-activated carbon (i.e., activated charcoal).  The charcoal 
absorbs radon that enters the canister via a screened opening.  After a determined expo-
sure period, the canister is sealed and the charcoal is analyzed using a HPGe gamma ray 
detector [Ger04].  Activated charcoal devices are preferable for short-term measurements 
(i.e., 1 to 7 days).  They are not practical for locations with high humidity because the 
charcoal can become saturated [Ger04]. 

The AlphaGUARD radon monitor is a continuous active radon sampling sensor.  It has 
an ionization chamber and uses an alpha spectroscopy to detect radon.  The two common 
isotopes of radon (i.e., 222Rn and 220Rn) are identified through their respective energies 
from the alpha decays [Swa04].  The signal generated from the alpha detection is converted 
to a digital output.  This output can be read by the AlphaGUARD,  dataloggers, or a PC.  
Ideal applications for the AlphaGUARD include radon exhalation, radon soil-gas, indoor, 
outdoor, in mining environments and in-water storage radon measurements. 

2.  AlphaGUARD Sensor Description
The AlphaGUARD is designed for long-term monitoring of radon gas concentration.  
It has a linear response from 2 to 2,000,000Bq/m3 (0.05 to 50,000 pCi/l).  See the next 
section and the AlphaGUARD brochure for more specifications.  

The AlphaGUARD measures radon concentration using 3D alpha spectroscopy.  There is a 
lag in the signal due to the time required for radon to diffuse into the ionization chamber.  
When exposed to a step change in radon concentration, the AlphaGUARD will reach 30% 
of the new value after 10 minutes, 70% after 20 minutes and 90% after 30 minutes.

The AlphaGUARD has an internal battery that will run it for approximately 10 days.  
The unit can be AC powered with an AC to 12 Vdc wall charger from Radio Shack or 
CSI pn15966.  Alternatively the AlphaGUARD may be DC powered using CSI pn14291.  
CSI pn14291 has a mating barrel connector on one end and red and black stripped and 
tinned leads to be connected to 12 Vdc and ground on the other end.

3.  Specifi cations
The best source of specifications on the AlphaGUARD is the manufacturer’s brochure 
which may be found at www.genitron.de/download/ag_fb_gb_01_a4.pdf.  A few specifi-
cations are listed below for your convenience.  Campbell Scientific recommends you refer 
to the manufacturer's brochure to verify critical specifications.

Measurement Range:  2  to  2,000,000 Bq/m3   (0.05  to  50,000 pC/liter)

Operating temperature:  +14°  to  +122°F (-10°  to  +50°C)

Operating pressure:  700  to  1100 mbar

Operating RH:   0  to  99% RH

Operating voltage:  12 Vdc (or 110 Vac with converter to 12 Vdc)

Current drain:   unknown (unit will run on internal battery for 10 days)

Figure 1.  Front view of the AlphaGUARD Professional Radon Monitor Model 
P30 (standard version).

4.  Installation
The AlphaGUARD must be switched on manually (On/Off switch).  Using the “Menu”, 
“Change”, and “Accept” buttons and the display, set the AlphaGUARD measuring cycle to 
10 minutes and the baud rate to 9600 baud.  The datalogger will time stamp the data but 
if it is important that the display indicate the correct time, set the AlphaGUARD clock 
using the serial interface software “DataEXPERT” provided with the unit.  

Th e AlphaGUARD user manual indicates the automatic background correction of the 
AlphaGUARD can only be executed at the beginning of a measuring series.  Th erefore with 
measurements lasting longer than six months, the set up of a new measuring series should be 
manually enforced, by turning the power switch OFF and then on again aft er every half year.  
Setting the system clock with DataEXPERT will also initiate a new measuring series.

5.  Wiring
The following wiring diagram is built for the program in Section 6 which is where the 
CR1000’s COM1 port (C1 & C2) will request and capture data from AlphaGUARD’s 
RS-232 port.  The color scheme indicated in the last column is specifi c to the two foot 
long “DTE” cable which is one of two cables that come in Campbell Scientifi c’s model 
SC110 (or pn16721).

TABLE 1:  AlphaGUARD to CR1000 Wiring

AlphaGUARD 
Function

AlphaGUARD
9-pin

CR1000
Terminal

CR1000
Function

CSI SC110 
DTE color

TxD 2 C2 RxD White

RxD 3 C1 TxD Brown

GND 5 GND GND Yellow

GND Shell GND Clear

Figure 2.  Above shows the AlphaGUARD connected to a CR1000 datalogger.

6.   Example Program for the CR1000

Th is CR1000 program is for the “standard version” of the AlphaGUARD, not the “overdrive” 
version.  It was tested on an AlphaGUARD with serial number 1635 EF.

'CR1000 Series Datalogger
'
'  Program for Alphaguard Radon detector
Public fpval
Public sign
Public  mantissa 
Public exp1
Public exponent
   
Public Radon,RadonErr,AirTemp,RelHum,BaroPress
Public Akkustd,KBeqstd,Totalstd
Public ASerial as string * 6,Aversion as string * 4
   
Public PTemp, batt_volt
' Public HexV(80) as String *2  ': for debugging
DataTable (Test,1,-1)
  DataInterval (0,15,Sec,10)
  Minimum (1,batt_volt,FP2,0,False)
  Sample (1,PTemp,FP2)
EndTable

Public Str3 As String * 80
Public CurrDate as String * 12
Public DataDate as String * 12
Public Atemp, Arh
Public start As Long,i, rtnbyte(2), array(129) As Long
Alias array(8) = year
Alias array(9) = month
Alias array(10)=days
Alias array(11)=hrs
Alias array(12)=minutes
Alias array(13)=seconds
Alias array(14)=mesyear
Alias array(15)=mesmonth
Alias array(16)=mesday
Alias array(17)=meshr
Alias array(18)=mesmin
Alias array(19)=messec
Alias array(20)=notused
Alias array(28)=cycletime
Alias array(35)=DataQA
Alias array(36)=sysQA

Function FLP2 (mant,expo) 

      sign = IIF(expo AND &H80,1.0,-1.0)
      mantissa = mant/256.0
     exp1  = (expo AND &H7F) - 64 
   exponent  = PWR(2,exp1) 
   FLP2 = (sign * mantissa) * exponent
 
EndFunction
Function FLP4 (mant1,mant2,mant3,expo) 

      sign = IIF(expo AND &H80,1.0,-1.0)
      mantissa = (mant3 + ((mant2*256)) + ((mant1*65536)))/16777216
     exp1  = (expo AND &H7F) - 64 
   exponent  = PWR(2,exp1) 
   FLP4 = (sign * mantissa) * exponent
 
EndFunction

BeginProg
  'Open COM1 at 9600 baud
  SerialOpen (Com1,9600,3,10,1000) ' format for binary, 8-bit, no parity.

  Scan (10,Sec,0,0)
    
   PanelTemp (PTemp,250)
  Battery (Batt_volt)

    If TimeIntoInterval (0,1,Min)
     start = 1
    endif
    
    If start Then
      ' Output string to query Alphaguard
      SerialFlush(com1)   ' Clean out buffers
      Str3 = CHR(2)
      SerialOutBlock (Com1,Str3,1)
      Delay (0,100,mSec)
      str3= CHR(3) + CHR(83) + CHR(3) + CHR(91)
      SerialOutBlock (Com1,Str3,4)
      
      ' Wait for return of data from Alphaguard
      Delay (0,4,Sec) 
      i=1
      rtnbyte(2)=SerialInChk(com1)

      While rtnbyte(2)
        i=i+1
        SerialInBlock(Com1,array(1),1)
        
        MoveBytes (array(i),3,array(1),0,1)

' following line useful for debugging        
'        hexV(i) = Hex(array(i))
        rtnbyte(2)=SerialInChk(com1)
      Wend
      
      start=0
      CurrDate = right("0" & Hex(year),2) & right("0" & Hex(month),2) & right("0" & Hex(days),2)
      CurrDate = CurrDate & right("0" & Hex(hrs),2) & right("0" & Hex(minutes),2) & right("0" & Hex(seconds),2)
      DataDate = right("0" & Hex(mesyear),2) & right("0" & Hex(mesmonth),2) & right("0" & Hex(mesday),2)
      DataDate = DataDate & right("0" & Hex(meshr),2) & right("0" & Hex(mesmin),2) & right("0" & Hex(messec),2)
      
      Radon = FLP2(array(32),array(31))
      RadonErr = FLP2(array(34),array(33))
      AirTemp = FLP2(array(40),array(39))
      RelHum = FLP2(array(46),array(45))
     
      BaroPress = FLP4(array(42),array(43),array(44),array(41))
      Akkustd = FLP4(array(48),array(49),array(50),array(47))
      KBeqstd = FLP4(array(52),array(53),array(54),array(51))
      Totalstd = FLP4(array(56),array(57),array(58),array(55))
     
      ASerial = CHR(array(60)) & CHR(array(61)) & CHR(array(62)) & CHR(array(63)) & CHR(array(64)) & CHR(array(65))
      AVersion = CHR(array(66)) & CHR(array(67)) & CHR(array(68)) & CHR(array(69))

    EndIf

  NextScan
EndProg

7.  Maintenance and Troubleshooting
See AlphaGUARD User Manual.

